

Exemplul 9.1

```

DECLARE
    x NUMBER(2) := &p_x;
    y NUMBER(2) := &p_y;
BEGIN
    DBMS_OUTPUT.PUT_LINE(x/y);
    DBMS_OUTPUT.PUT_LINE(x);
    DBMS_OUTPUT.PUT_LINE(y);
EXCEPTION
    WHEN ZERO_DIVIDE THEN
        DBMS_OUTPUT.PUT_LINE('Codul erorii: ' || SQLCODE);
        DBMS_OUTPUT.PUT_LINE('Mesajul erorii: ' || SQLERRM);
END;
/

```

Exemplul 9.2

```

DECLARE
    v_nr    NUMBER := &p_nr;
    v_den   categorii.denumire%TYPE;
BEGIN
    SELECT denumire INTO v_den
    FROM categorii
    WHERE id_categorie = (SELECT id_categorie
                           FROM produse
                           GROUP BY id_categorie
                           HAVING COUNT(*) = v_nr);
    DBMS_OUTPUT.PUT_LINE('Categoria gasita: ' || v_den);
EXCEPTION
    WHEN NO_DATA_FOUND THEN
        DBMS_OUTPUT.PUT_LINE('Nu exista categorii cu ' ||
                             ' numarul specificat de produse');
    WHEN TOO_MANY_ROWS THEN
        DBMS_OUTPUT.PUT_LINE('Există mai multe categorii ' ||
                             ' cu numarul specificat de produse');
    WHEN OTHERS THEN
        DBMS_OUTPUT.PUT_LINE('Codul erorii: ' || SQLCODE);
        DBMS_OUTPUT.PUT_LINE('Mesajul erorii: ' || SQLERRM);
END;
/

```

Exemplul 9.3

```

DECLARE
    x NUMBER(2) := &p_x;
    y NUMBER(2) := &p_y;
BEGIN
    IF x<0 OR y<0 THEN
        -- declansare explicita
        RAISE INVALID_NUMBER;
    END IF;
    -- declansare implicita
    DBMS_OUTPUT.PUT_LINE(TO_CHAR(x/y,'999.99.99'));

EXCEPTION
    WHEN ZERO_DIVIDE THEN
        DBMS_OUTPUT.PUT_LINE('Impartirea la 0 nu ' ||
            ' este permisa');
    WHEN INVALID_NUMBER THEN
        -- declansare implicita
        DBMS_OUTPUT.PUT_LINE('Nu sunt permise valori ' ||
            ' negative!');
    WHEN OTHERS THEN
        -- declansare implicita
        DBMS_OUTPUT.PUT_LINE ('Codul erorii: ' ||SQLCODE);
        DBMS_OUTPUT.PUT_LINE ('Mesajul erorii: ' ||SQLERRM);
END;
/

```

Exemplul 9.4

```

DECLARE
    exista_produse EXCEPTION;
    PRAGMA EXCEPTION_INIT(exista_produse,-2292);
    v_denumire categorii.denumire%TYPE := '&p_denumire';

BEGIN
    DELETE FROM categorii WHERE denumire = v_denumire;
    IF SQL%NOTFOUND THEN
        RAISE NO_DATA_FOUND;
    END IF;

EXCEPTION
    WHEN exista_produse THEN
        DBMS_OUTPUT.PUT_LINE ('Nu puteti sterge categoria ' ||
            v_denumire || ' deoarece contine produse');
    WHEN NO_DATA_FOUND THEN
        DBMS_OUTPUT.PUT_LINE ('Ati introdus date eronate');
END;
/
-- incalcarea constrangerii de cheie externa determina
-- eroarea cu codul -2292

```

Exemplul 9.5

```

DECLARE
    exceptie          EXCEPTION;
    cantitate_totala NUMBER;
    v_id              produse.id_produs%TYPE;
    v_denumire        produse.denumire%TYPE
                      := '&p_denumire';
BEGIN
    SELECT id_produs INTO v_id
    FROM produse
    WHERE denumire = v_denumire;

    SELECT SUM(cantitate) INTO cantitate_totala
    FROM facturi_produse
    WHERE id_produs = v_id;

    IF cantitate_totala IS NULL THEN
        RAISE exceptie;
    END IF;

    DBMS_OUTPUT.PUT_LINE('Produs '||v_denumire||
                          ' - Numar bucati vandute = '||cantitate_totala);
EXCEPTION
    WHEN TOO_MANY_ROWS THEN
        DBMS_OUTPUT.PUT_LINE('Există mai multe produse cu
                              aceeași denumire!');
    WHEN NO_DATA_FOUND THEN
        DBMS_OUTPUT.PUT_LINE('Nu există produsul cu
                              denumirea specificată!');
    WHEN exceptie THEN
        DBMS_OUTPUT.PUT_LINE('Produsul nu are
                              vanzari înregistrate');
    WHEN others THEN
        DBMS_OUTPUT.PUT_LINE('Alte eroare');
END;
/

```